UNITED STATES DEPARTMENT of the INTERIOR

GEOLOGICAL SURVEY

Solutions for Apparent Ages and U-Pb Isotope Ratios of Young Systems without Initial Secular Equilibrium: A Computer Program in Hewlett-Packard BASIC

Kenneth R. Ludwig
Denver, Colorado 80225

Open-File Report 79-722

1979

Solutions for Apparent Ages and U-Pb Isotope Ratios of Young Systems without Initial Secular Equilibrium: A Computer Program in Hewlett-Packard BASIC

By Kenneth R. Ludwig

INTRODUCTION

Most U-Pb isotope calculations made by geochronologists assume that the systems of interest have always been in secular radioactive equilibrium, so that the effects of growth and decay of the several long-lived radioactive daughters of ^{238}U and ^{235}U can be ignored. Although this approach is reasonable for relatively old (> 30 m.y.) systems, the inaccuracies introduced by these assumptions are quite substantial when very young (< 1-2 m.y.) systems are encountered (Ludwig, 1977; Ludwig and others, 1977).

The following computer program permits solution of the Bateman Equations (which describe the relative abundances of isotopes in a decay-chain) in three different ways:

$^{206}Pb/^{238}U$, $^{207}Pb/^{235}U$, and $^{207}Pb/^{206}Pb$ ratios can be calculated in table form for a given series of ages and given initial radioactive-daughter abundances (ratio-from-age); or the $^{206}Pb/^{238}U$ and $^{207}Pb/^{235}U$ apparent ages can be calculated from the isotope ratios (age-from-ratio). The third option permits calculation of the initial $^{234}U/^{238}U$ activity ratio necessary for a closed system of given age and given observed $^{207}Pb/^{206}Pb$ ratio.

All isotopes in the decay chains with half-lives > 1 yr. are included in the calculations, using the following half-lives:

Isotope	Half-life
^{238}U	4468 m.y.
^{234}U	0.247 m.y.
^{230}Th	0.077 m.y.
^{226}Ra	1600 yr.
^{210}Pb	22.3 yr.
^{235}U	703.8 m.y.
^{231}Pa	0.03250 m.y.
^{227}Ac	21.8 yr.

The user may specify estimated initial activity ratios of $^{234}U/^{238}U$, $^{230}Th/^{238}U$, and $^{231}Pa/^{235}U$, but initial ^{226}Ra, ^{210}Pb, and ^{227}Ac abundances are assumed to be negligible.

USE OF THE PROGRAM

The program was written for a Hewlett-Packard model 9831 desktop computer and requires about 2500 words (5000 bytes) of memory. With little or no modification, the program should also function in Hewlett-Packard models 9830, 9835, and 9845 computers. Use of the program is essentially self-explanatory. Sample outputs are given in figures 1 and 2. The program itself is given in the appendix.

The ratio-from age calculations are done by a straight forward solution of the Bateman Equations (Kirby, 1973). The age from ratio and initial $^{234}U/^{238}U$ calculations are done by a rapidly converging iterative method involving calculation of the derivatives of the age-ratio and $^{207}Pb/^{206}Pb$-$^{234}U/^{238}U$ functions for each trial solution. In some extreme cases (e.g., an initial activity ratio for $^{234}U/^{238}U$ of 100), the program may fail to find a solution, though for geologically and analytically possible cases this is unlikely.

I will be happy to copy the program onto HP-9831 or 9830 tape cassettes for any interested persons.

REFERENCES

Kirby, H. W., 1973, Nuclear properties and genetic relationships of the naturally occurring radioactive series: Atomic Energy Comm. Research and Development Report MLM-2036, p. 25-43.

Ludwig, K. R., 1977, Effect of initial radioactive-daughter disequilibrium on U-Pb isotope apparent ages of Young minerals: U.S. Geological Survey Journal of Research, v. 5, no. 6, p. 663-667.

Ludwig, K. R., Szabo, B. J., and Granger, H. G., 1977, Pleistocene apparent ages by U-Pb isotope and U-series methods for uranium ore in Dakota Sandstone near Gallup, New Mexico: U.S. Geological Survey Journal of Research, v. 5, no. 6, p. 664-672.

FIGURE 1. -- Sample of option 1 printout.

```
RATIOS FROM AGES (1), AGES FROM RATIOS (2), OR INITIAL 234/238
FROM CONCORDANT Pb/U AGES AND 207/206 (3)?


INITIAL ACTIVITY RATIOS: 234/238= 1.8  230/238= 0.1  231/235= 0

AGE(M.Y.)          206/238          207/235          207/206

    0.020      6.2787E-07       3.6516E-06       0.042180
    0.040      2.0718E-06       1.2375E-05       0.045073
    0.060      4.1567E-06       2.5736E-05       0.044905
    0.080      6.7569E-06       4.0972E-05       0.043978
    0.100      9.7686E-06       5.7757E-05       0.042882


INITIAL ACTIVITY RATIOS: 234/238= 0.6  230/238= 2    231/235= 0.5

AGE(M.Y.)          206/238          207/235          207/206

    1.000      1.5185E-04       9.6224E-04       0.045958
    2.000      3.0519E-04       1.9486E-03       0.046306
    3.000      4.6028E-04       2.9359E-03       0.046261
    4.000      6.1549E-04       3.9241E-03       0.046241
    5.000      7.7073E-04       4.9134E-03       0.046236
    6.000      9.2600E-04       5.9036E-03       0.046239
    7.000      1.0813E-03       6.8948E-03       0.046246
    8.000      1.2366E-03       7.8870E-03       0.046257
    9.000      1.3919E-03       8.8801E-03       0.046270
   10.000      1.5473E-03       9.8743E-03       0.046283
```

FIGURE 2. -- Samples of option 2 printout.

```
RATIOS FROM AGES (1), AGES FROM RATIOS (2), OR INITIAL 234/238
FROM CONCORDANT Pb/U AGES AND 207/206 (3)?

INITIAL ACTIVITY RATIOS: 234/238= 1.8  230/238= 0.1  231/235= 0

206/238= 2.0718E-06          AGE=   0.0400 M.Y.

207/235= 1.2875E-05          AGE=   0.0400 M.Y.
-------------------------------------------------------------------

INITIAL ACTIVITY RATIOS: 234/238= 0.6  230/238= 2    231/235= 0.5

206/238= 1.3919E-03          AGE=   8.9997 M.Y.

207/235= 8.8801E-03          AGE=   9.0000 M.Y.
-------------------------------------------------------------------
```

FIGURE 3. -- Sample of option 3 printout.

```
RATIOS FROM AGES (1), AGES FROM RATIOS (2), OR INITIAL 234/238
FROM CONCORDANT Pb/U AGES AND 207/206 (3)?

-------------------------------------------------------------------------
CALCULATION OF INITIAL 234/238 FOR A GIVEN 207/206

AGE (M.Y.)      207/206         (230/238)0      (231/235)0

 0.06            0.044905        0.1             0

INITIAL 234/238 MUST BE  1.800 FOR A CLOSED-SYSTEM
-------------------------------------------------------------------------

AGE (M.Y.)      207/206         (230/238)0      (231/235)0

 9               0.04627         2               0.5

INITIAL 234/238 MUST BE  0.600 FOR A CLOSED-SYSTEM
-------------------------------------------------------------------------

AGE (M.Y.)      207/206         (230/238)0      (231/235)0

 2.30000E-03     0.17941         0.2             0.7

INITIAL 234/238 MUST BE 41.336 FOR A CLOSED-SYSTEM
-------------------------------------------------------------------------

AGE (M.Y.)      207/206         (230/238)0      (231/235)0
```

APPENDIX -- LIST OF PROGRAM

```
10 REM SOLUTIONS TO BATEMAN'S EQUATIONS FOR THE U-Pb ISOTOPE SYSTEM
20 REM FOR HP-9831/9830 DESKTOP COMPUTERS.  K.R. LUDWIG, 2/79
30 DIM L5[5],M5[3],C[5],D[5],E[5],A5[3],F[5]
40 REM DECAY CONSTANTS FOR 238U,234U,230Th AND 210Pb
50 DATA 1.55125E-04,2.8263,9.0019,433.22,3.1083E+04
60 REM DECAY CONSTANTS FOR 235U, 231Pa, AND 227Ac
70 DATA 9.8485E-04,21.328,3.184E+04
80 READ L[1],L[2],L[3],L[4],L[5],M[1],M[2],M[3]
90 IF STAT6>256 THEN 120
100 P0=2
110 GOTO 160
120 P0=2+4*(STAT6>287)
160 STDPP P0
170 PRINT LIN1"RATIOS FROM AGES (1), AGES FROM RATIOS (2), OR INITIAL 234/238"
180 PRINT "FROM CONCORDANT Pb/U AGES AND 207/206 (3)?"LIN3
190 INPUT F
200 GOTO 390
210 END
220 DEF FNA(I)
230 L1=L2=1
240 FOR J=N0 TO 5
250 IF I=J THEN 280
260 L1=-L1*L[J]
270 L2=L2*(L[I]-L[J])
280 NEXT J
290 RETURN -L1/L2
300 DEF FNB(I)
310 M1=M2=1
320 FOR J=N0 TO 3
330 IF I=J THEN 360
340 M1=-M1*M[J]
350 M2=M2*(M[I]-M[J])
360 NEXT J
370 RETURN -M1/M2
380 REM BATEMAN COEFFICIENTS CALCULATION
390 FOR I=1 TO 5
400 N0=1
410 C[I]=FNA(I)
420 IF I=1 THEN 480
430 N0=2
440 D[I]=FNA(I)
450 IF I<3 THEN 480
```

```
460 N0=3
470 E[I]=FNAI
480 NEXT I
490 FOR I=1 TO 3
500 N0=1
510 P[I]=FNBI
520 IF I=1 THEN 550
530 N0=2
540 Q[I]=FNBI
550 NEXT I
560 GOTO F OF 1330,1540,1830
570 GOTO 190
580 DEF FNC(T)
590 REM 206/238 CALCULATION
600 R1=R2=R3=0
610 F2=EXP(L[1]*T)
620 FOR I=1 TO 5
630 F[I]=-T*L[I]
640 Z1=-180-20*(I#5)
650 F[I]=F[I]*(F[I] >= Z1)+Z1*(F[I]<Z1)
660 F[I]=EXPF[I]
670 R1=R1+C[I]*F[I]
680 IF I=1 THEN 720
690 R2=R2+D[I]*F[I]
700 IF I <= 2 THEN 720
710 R3=R3+E[I]*F[I]
720 NEXT I
730 RETURN F2*(1+R1+L[1]*A[1]*(1+R2)/L[2]+L[1]*A[2]*(1+R3)/L[3])
740 DEF FND(T)
750 REM SLOPE OF 206/238 - T CURVE
760 R1=R2=R3=0
770 F2=L[1]*EXP(L[1]*T)
780 FOR I=1 TO 5
790 F[I]=(L[1]-L[I])*T
800 Z1=-180-20*(I#5)
810 F[I]=F[I]*(F[I] >= Z1)+Z1*(F[I]<Z1)
820 F[I]=(L[1]-L[I])*EXPF[I]
830 R1=R1+C[I]*F[I]
840 IF I=1 THEN 860
850 R2=R2+D[I]*F[I]
860 IF I <= 2 THEN 880
870 R3=R3+E[I]*F[I]
880 NEXT I
890 RETURN F2+R1+(A[1]*L[1]/L[2])*(F2+R2)+(A[2]*L[1]/L[3])*(F2+R3)
900 DEF FNE(T)
910 REM 207/235 CALCULATION
920 S1=S2=0
930 F2=EXP(M[1]*T)
940 FOR I=1 TO 3
950 F[I]=-T*M[I]
```

```
960 Z1=-160
970 F[I]=F[I]*(F[I] >= Z1)+Z1*(F[I]<Z1)
980 F[I]=EXPF[I]
990 S1=S1+P[I]*F[I]
1000 IF I=1 THEN 1020
1010 S2=S2+Q[I]*F[I]
1020 NEXT I
1030 RETURN F2*(1+S1+M[1]*A[3]*(1+S2)/M[2])
1040 DEF FNF(T)
1050 REM SLOPE OF 207/235 - T CURVE
1060 S1=S1=0
1070 F2=M[1]*EXP(M[1]*T)
1080 FOR I=1 TO 3
1090 F[I]=(M[1]-M[I]*T)
1100 Z1=-160
1110 F[I]=F[I]*(F[I] >= Z1)+Z1*(F[I]<Z1)
1120 F[I]=(M[1]-M[I])*EXPF[I]
1130 S1=S1+P[I]*F[I]
1140 IF I=1 THEN 1160
1150 S2=S2+Q[I]*F[I]
1160 NEXT I
1170 RETURN F2+S1+(A[3]*M[1]/M[2])*(F2+S2)
1180 END
1190 DEF FNG(T)
1200 REM SLOPE OF 207/206 - (234/238)0 CURVE
1210 R5=0
1220 F2=EXP(L[1]*T)
1230 FOR I=2 TO 5
1240 F[I]=-T*L[I]
1250 Z1=-180-20*(I#5)
1260 F[I]=F[I]*(F[I] >= Z1)+Z1*(F[I]<Z1)
1270 F[I]=EXPF[I]
1280 R5=R5+D[I]*F[I]
1290 NEXT I
1300 RETURN F2*L[1]*(1+R5)/L[2]
1310 END
1320 REM RATIOS FROM AGES
1330 GOSUB 1350
1340 GOTO 1410
1350 DISP "INITIAL 234/238,230/238,231/235";
1360 INPUT A[1],A[2],A[3]
1370 STANDARD
1380 PRINT "INITIAL ACTIVITY RATIOS: 234/238="A[1]"230/238="A[2]"231/235="A[3]
1390 PRINT LIN1
1400 RETURN
1410 DISP "START, STOP, INTERVAL (M.Y.)";
1420 INPUT T1,T2,I2
1430 PRINT "AGE(M.Y.)"TAB18"206/238"TAB37"207/235"TAB56"207/206"LIN2
1440 FOR T=T1 TO T2 STEP I2
1450 R=FNCT
```

```
1460 S=FNET
1470 FORMAT F7.3,9X,E11.4,8X,E11.4,8X,F10.6
1480 WRITE (P0,1470)T,F,S,S/(P*137.88)
1490 NEXT T
1500 PRINT LIN3
1510 GOTO 1330
1520 END
1530 PEM AGES FROM RATIOS
1540 GOSUB 1350
1550 DISP "206/238,  207/235";
1560 INPUT Y,Y1
1570 IF Y THEN 1600
1580 T8=0
1590 GOTO 1670
1600 T=LOG(1+Y)/L[1]
1610 T2=T+(Y-FNCT)/FNDT
1620 DISP T2
1630 IF ABS(T-T2)<T2*1E-04 THEN 1660
1640 T=T2
1650 GOTO 1610
1660 T8=T
1670 IF Y1 THEN 1700
1680 T=0
1690 GOTO 1760
1700 T=LOG(1+Y1)/M[1]
1710 T3=T+(Y1-FNET)/FNFT
1720 DISP T2;T3
1730 IF ABS(T-T3)<T3*1E-04 THEN 1760
1740 T=T3
1750 GOTO 1710
1760 FORMAT "206/238=",E11.4,10X,"AGE=",F9.4," M.Y.",/,/
1770 FORMAT "207/235=",E11.4,10X,"AGE=",F9.4," M.Y.",/,72"-",/,/
1780 WRITE (P0,1760)Y,T8
1790 WRITE (P0,1770)Y1,T
1800 GOTO 1540
1810 END
1820 PEM (234/238)0 FROM 207/206
1830 FORMAT /,72"-",/,"CALCULATION OF INITIAL 234/238 FOR A GIVEN 207/206",/,/
1840 WRITE (P0,1830)
1850 PRINT "AGE (M.Y.)","207/206","(230/238)0","(231/235)0"LIN1
1860 INPUT T,Y,A[2],A[3]
1870 PRINT T,Y,A[2],A[3],LIN1
1880 A[1]=A1=1
1890 C=FNCT
1900 E=FNET
1910 Y1=E/(137.88*C)
1920 F1=-FNCT*E/(137.88*C*C)
1930 A1=A[1]+(Y-Y1)/F1
1940 DISP A1
1950 IF ABS(A[1]-A1)<1E-04 THEN 1980
```

```
1960 A[1]=A1
1970 GOTO 1890
1980 FORMAT "INITIAL 234/238 MUST BE",F7.3," FOR A CLOSED-SYSTEM",/,72"-",/,/
1990 WRITE (P0,1980)A1
2000 GOTO 1850
2010 END
```